Statistical Physics II.
© Daniel Alonso. IUdEA

Fermi ideal gas. Equation of state

```
f[q_, z_] := Sum[(-1)^(k+1) z^k / k^q, {k, 1, 5}]
```

Case 1.- High temperatures and low densities

- **The Fermi case:** $x = \lambda^3 / v$, **v=Volumen per particle**

```
S = f[3 / 2, z] // Expand
```

$$z - \frac{z^2}{2\sqrt{2}} + \frac{z^3}{3\sqrt{3}} - \frac{z^4}{8} + \frac{z^5}{5\sqrt{5}}$$

```
InvS = InverseSeries[S + O[z]^6, z]
```

$$z + \frac{z^2}{2\sqrt{2}} + \frac{1}{36}\left(9 - 4\sqrt{3}\right) z^3 + \frac{1}{288}\left(36 + 45\sqrt{2} - 40\sqrt{6}\right) z^4 + \frac{\left(2375 + 1350\sqrt{2} - 2100\sqrt{3} - 288\sqrt{5}\right) z^5}{7200} + O[z]^6$$

```
y = Normal[InvS] /. {z → x}
```

$$x + \frac{x^2}{2\sqrt{2}} + \frac{1}{36}\left(9 - 4\sqrt{3}\right) x^3 + \frac{1}{288}\left(36 + 45\sqrt{2} - 40\sqrt{6}\right) x^4 + \frac{\left(2375 + 1350\sqrt{2} - 2100\sqrt{3} - 288\sqrt{5}\right) x^5}{7200}$$

```
Series[Expand[f[5 / 2, y] / x], {x, 0, 3}]
```

$$1 + \frac{x}{4\sqrt{2}} + \left(\frac{1}{8} - \frac{2}{9\sqrt{3}}\right) x^2 + \frac{1}{192}\left(18 + 15\sqrt{2} - 16\sqrt{6}\right) x^3 + O[x]^4$$

- Then using the relation $Pv/K_B T) = x^{-1} f[5/2, x]$ it follows;

$$Pv/K_B T = 1 + (\lambda^3/v)/(2^{5/2}) + \left(\frac{1}{8} - \frac{2}{9\sqrt{3}}\right)(\lambda^3/v)^2 + \ldots$$

Case 2.- Low Temperatures and high densities

- We consider the integral version of $f_{3/2}$ and develop the Sommenrfeld expansion

```
Inte[n_] := Integrate[u^n Exp[u]/(Exp[u] + 1)^2, {u, -Infinity, Infinity}]
```

```
Intable = Table[Inte[n], {n, 0, 10}]
```

$$\left\{1, 0, \frac{\pi^2}{3}, 0, \frac{7\pi^4}{15}, 0, \frac{31\pi^6}{21}, 0, \frac{127\pi^8}{15}, 0, \frac{2555\pi^{10}}{33}\right\}$$

```
Intable[[3]]
```

$$\frac{\pi^2}{3}$$

- The result is:

```
f32Asymp[z_] :=
  4/(3 Sqrt[π]) (Intable[[1]] Log[z]^(3/2) + 3/8 Intable[[3]] Log[z]^(-1/2))

f32[z_] := 4/Sqrt[π] NIntegrate[x^2/(z^-1 Exp[x^2] + 1), {x, 0, Infinity}]
```

We compare the asymptotic formula and the result of a numerical integration

```
zmin = 0.5; zmax = 100;
g2 = Plot[f32Asymp[z], {z, zmin, zmax}];
g1 = Plot[f32[z], {z, zmin, zmax}];
Show[g2, g1]
```



- **Analysis of the integral** $\int_{-\infty}^{-\nu} \frac{e^u u^n}{(e^u+1)^2} du$

```
(* We study the simplest case *)

g[ν_] := NIntegrate[Exp[u]/(Exp[u] + 1)^2, {u, -Infinity, -ν}]
```

$(* \int_{-\infty}^{-\nu} \frac{e^u u^n}{(e^u+1)^2} du *)$

```
(* We observe a decay,
the question is how it decays the function g *)
Plot[g[x], {x, 1, 10}, AxesLabel → {ν, "g[ν]"}]
```



```
data = Table[{x, Log[g[x]]}, {x, 10, 100, 1}]
ListPlot[data, AxesLabel → {ν, "Log[g[ν]]"}]
(* We plot the data obtained bynumerical integration *)
Fit[data, {1, x}, x] (* Linear fitting *)
(* Observe that the slope obtained is, in this case -1.! *)
```

```
{{10, -10.}, {11, -11.}, {12, -12.}, {13, -13.}, {14, -14.}, {15, -15.}, {16, -16.},
 {17, -17.}, {18, -18.}, {19, -19.}, {20, -20.}, {21, -21.}, {22, -22.}, {23, -23.},
 {24, -24.}, {25, -25.}, {26, -26.}, {27, -27.}, {28, -28.}, {29, -29.}, {30, -30.},
 {31, -31.}, {32, -32.}, {33, -33.}, {34, -34.}, {35, -35.}, {36, -36.}, {37, -37.},
 {38, -38.}, {39, -39.}, {40, -40.}, {41, -41.}, {42, -42.}, {43, -43.}, {44, -44.},
 {45, -45.}, {46, -46.}, {47, -47.}, {48, -48.}, {49, -49.}, {50, -50.}, {51, -51.},
 {52, -52.}, {53, -53.}, {54, -54.}, {55, -55.}, {56, -56.}, {57, -57.}, {58, -58.},
 {59, -59.}, {60, -60.}, {61, -61.}, {62, -62.}, {63, -63.}, {64, -64.}, {65, -65.},
 {66, -66.}, {67, -67.}, {68, -68.}, {69, -69.}, {70, -70.}, {71, -71.}, {72, -72.},
 {73, -73.}, {74, -74.}, {75, -75.}, {76, -76.}, {77, -77.}, {78, -78.}, {79, -79.},
 {80, -80.}, {81, -81.}, {82, -82.}, {83, -83.}, {84, -84.}, {85, -85.}, {86, -86.},
 {87, -87.}, {88, -88.}, {89, -89.}, {90, -90.}, {91, -91.}, {92, -92.}, {93, -93.},
 {94, -94.}, {95, -95.}, {96, -96.}, {97, -97.}, {98, -98.}, {99, -99.}, {100, -100.}}
```



$-3.58352\times10^{-6} - 1.\,x$

As expected the slope is -1 !

Ideal Fermi gas at low temperatures: Average occupation number

```
np[β_, ϵ_] := (1 + Exp[β (ϵ - 1)])^-1;
```

```
(* We introduce here few commands:
  1.- How to plot a function and
    define its window range (command PlotRange)
    2.- How to use Manipulate "something",
in this case a Plot *)

Manipulate[Plot[np[β, ϵ], {ϵ, 0, 3},
  PlotRange → {{0, 3}, {0, 1.1}}], {β, 1, 200}]
```